# Maintainer: Balló György <ballogyor+arch at gmail dot com>
# Contributor: Dimitris Kiziridis <ragouel at outlook dot com>

pkgname=minder
pkgver=1.15.1
pkgrel=1
pkgdesc='Mind-mapping application designed for Pantheon'
arch=('loong64' 'x86_64')
url='https://github.com/phase1geo/Minder'
license=('GPL3')
depends=('discount' 'granite' 'gtksourceview4' 'libarchive' 'libhandy')
makedepends=('meson' 'vala')
source=("https://github.com/phase1geo/Minder/archive/$pkgver/Minder-$pkgver.tar.gz")
sha256sums=('907caab7bac62c356f3e5e15b5cfa001880d12d59f3b543756a2ea955f29b0a6')

build() {
  arch-meson Minder-$pkgver build
  meson compile -C build
}

package() {
  meson install -C build --destdir "$pkgdir"
  ln -s com.github.phase1geo.minder "$pkgdir/usr/bin/$pkgname"
}
